Quadriken 3d Hyperbolischer Zylinder

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Juni 07 Update 30.06.07
Web: http://haftendorn.uni-lueneburg.de www.mathematik-verstehen.de

###

Rückwärts aufgestellt, siehe Datei Konstruktion-rückwärts

```
p:=matrix([x,y,z]):pt:=linalg::transpose(p)
```

$\begin{pmatrix} x & y & z \end{pmatrix}$

```
quadrik:=matrix([[8*z - 12*y - 12*x - 1/2*x^2 - 1/2*y^2
- x*y + 2*x*z + 2*y*z + 40]])
```

$$\left(8\cdot z - y\cdot 12 - x\cdot 12 - \frac{x^2}{2} - \frac{y^2}{2} - x\cdot y + 2\cdot x\cdot z + 2\cdot y\cdot z + 40 \right)$$

```
quadrikp:=plot::Implicit3d(quadrik[1],x=-20..20,y=-20..20, z=-20..20,FillColor=[0,1,0,0.2]
):
plot(quadrikp)
```



A passend aufstellen

```
A:=matrix([[-1/2,-1/2,1],[-1/2,-1/2,1],[1,1,0]]);
a:=matrix([-12,-12,8]): at:=linalg::transpose(a);
d:=40;
```

$$\begin{pmatrix} -\frac{1}{2} & -\frac{1}{2} & 1 \\ -\frac{1}{2} & -\frac{1}{2} & 1 \\ 1 & 1 & 0 \end{pmatrix}$$

$\begin{pmatrix} -12 & -12 & 8 \end{pmatrix}$

40


```
expand(pt*A*p+at*p+d);  //Probe, ob man A,a und d
richtig hat
```

```
richtig hat
%-quadrik
```

$$\left(8\cdot z-y\cdot 12-x\cdot 12-\frac{x^2}{2}-\frac{y^2}{2}-x\cdot y+2\cdot x\cdot z+2\cdot y\cdot z+40\right)$$

$$\left(0\right)$$

hier muss 0 herauskommen

Hauptachtsentransformation

```
E3:=matrix([[1,0,0],[0,1,0],[0,0,1]])
```

$$\begin{pmatrix}1&0&0\\0&1&0\\0&0&1\end{pmatrix}$$

```
evli:=linalg::eigenvectors(A) //Probe, was MuPAD liefert
```

$$\left[\left[0,\,1,\,\left[\begin{pmatrix}-1\\1\\0\end{pmatrix}\right]\right],\,\left[-2,\,1,\,\left[\begin{pmatrix}-1\\-1\\1\end{pmatrix}\right]\right],\,\left[1,\,1,\,\left[\begin{pmatrix}\frac{1}{2}\\\frac{1}{2}\\1\end{pmatrix}\right]\right]\right]$$

```
ew1 :=evli[1][1];  ew2 :=evli[2][1]; ew3 :=evli[3][1];
ev1:=evli[1][3][1];
ev2:=evli[2][3][1];
ev3:=evli[3][3][1];
```

$$0$$

$$-2$$

$$1$$

$$\begin{pmatrix}-1\\1\\0\end{pmatrix}$$

$$\begin{pmatrix}-1\\-1\\1\end{pmatrix}$$

$$\begin{pmatrix}\frac{1}{2}\\\frac{1}{2}\\1\end{pmatrix}$$

```
ev1n:=linalg::normalize(ev1):
ev2n:=linalg::normalize(ev2):
ev3n:=linalg::normalize(ev3):
P:=ev1n.ev2n.ev3n: Pt:=linalg::transpose(P);
```


$$\begin{pmatrix}-\frac{\sqrt{2}}{2}&\frac{\sqrt{2}}{2}&0\\-\frac{\sqrt{3}}{\ldots}&-\frac{\sqrt{3}}{\ldots}&\frac{\sqrt{3}}{\ldots}\end{pmatrix}$$

$$\begin{pmatrix} -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \\ -\frac{\sqrt{3}}{3} & -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \\ \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{3} \end{pmatrix}$$

Vektorschreibweise für die Abbildung $\vec{p} = P\,\vec{p}\,'$ und die Quadrikgleichungen, die sich durch Einsetzen ergeben:

$$Q: \quad \vec{p}^{\,T} A\,\vec{p} + \vec{a}^{\,T}\vec{p} + d = 0$$

$$Q': \quad \vec{p}\,'^{T} P^T A\,P\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

$$Q': \quad \vec{p}\,'^{T} D_{EW}\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

mit $D_{EW} = \begin{pmatrix} \lambda_1 & 0 \\ 0 & \lambda_2 \end{pmatrix}$

```
quastrich:=Simplify(pt*Pt*A*P*p+at*P*p+d)
```

$$\left(\frac{32\cdot\sqrt{3}\cdot y}{3} - y^2\cdot 2 + z^2 - \frac{\sqrt{6}\cdot z\cdot 4}{3} + 40\right)$$

```
quastrichp:=plot::Implicit3d(quastrich[1],x=-20..20,y=-2
0..20, z=-20..20,Filled=FALSE):
plot(quastrichp):
```



3

```
quastrich
```

$$\left(\frac{32\cdot\sqrt{3}\cdot y}{3} - y^2\cdot 2 + z^2 - \frac{\sqrt{6}\cdot z\cdot 4}{3} + 40\right)$$

$$\left(\frac{32\cdot\sqrt{3}\cdot y}{3} - y^2\cdot 2 + z^2 - \frac{\sqrt{6}\cdot z\cdot 4}{3} + 40\right)$$

Die Arbeitsweise ist dieselbe wie bei der Herstellung der Scheitelform einer Parabel.
Hier durch Hinsehen:

```
//xterm:=hold(2*(x+7/2*sqrt(2))^2);expand(xterm);
yterm:=hold(-2*(y-8/3*sqrt(3))^2);expand(yterm);
zterm:=hold((z-2/3*sqrt(6))^2);expand(zterm);
```

$$-2\cdot\left(y - \frac{8\cdot\sqrt{3}}{3}\right)^2$$

$$-y^2\cdot 2 + \frac{32\cdot\sqrt{3}\cdot y}{3} - \frac{128}{3}$$

$$\left(z - \frac{2\cdot\sqrt{6}}{3}\right)^2$$

$$z^2 - \frac{\sqrt{6}\cdot z\cdot 4}{3} + \frac{8}{3}$$

Also

```
quastrichK:=yterm+zterm+80   // i.a. mit xterm
```

$$\left(z - \frac{\sqrt{6}\cdot 2}{3}\right)^2 - \left(y - \frac{\sqrt{3}\cdot 8}{3}\right)^2\cdot 2 + 80$$

```
quastrich-expand(quastrichK)
```

$$(0)$$

hier muss 0 herauskommen

Letzter Teil der Hauptachsentransformation ist die Translation t

```
t:=matrix([s, -8/3*sqrt(3),-2/3*sqrt(6)]);
tt:=linalg::transpose(t):
```

$$\begin{pmatrix} s \\ -\frac{8\cdot\sqrt{3}}{3} \\ -\frac{2\cdot\sqrt{6}}{3} \end{pmatrix}$$

$\vec{p}\,'' = \vec{p}\,' + \vec{t}$ also $\vec{p}\,' = \vec{p}\,'' - \vec{t}$ Das ergibt:

```
quaH:=Simplify(expand((pt-tt)*Pt*A*P*(p-t)+at*P*(p-t)))+
40;
quadrikH:=ew1*x^2+ew2*y^2+ew3*z^2+80
```


$$\left(-y^2\cdot 2 + z^2 + 80\right)$$

$$\left(-y^2\cdot 2+z^2+80\right)$$

$$-y^2\cdot 2+z^2+80$$

Angabe der Gleichung in der üblichen Form:

```
hold(y^2/40-z^2/80=1)
```

$$\frac{y^2}{40}-\frac{z^2}{80}=1$$

```
quadrikHp:=plot::Implicit3d(quadrikH,x=-20..20,y=-20..20
, z=-20..20
        ):
plot(%)
```



Bestimmung des ursprünglichen Mittelpunktes:

$$\vec{m}''=\begin{pmatrix}0\\0\end{pmatrix},\quad \vec{m}'=-\vec{t},\quad \vec{m}=P(-\vec{t})$$

```
ms:=Simplify(P*(-t))
```

$$\begin{pmatrix}\frac{\sqrt{2}\cdot s}{2}-2\\ -\frac{\sqrt{2}\cdot s}{2}-2\\ 4\end{pmatrix}$$

Auf dieser Geraden liegt der alte Mittelpunkt.

```
m:=ms|s=0;
```

$$\begin{pmatrix}-2\\-2\\4\end{pmatrix}$$

Den bei der Rückwärtsrechnung verwendeten Mittelpunkt erhält man für

Den bei der Rückwärtsrechnung verwendeten Mittelpunkt erhält man für

```
solve({ms[1]=-1,ms[2]=-3,ms[3]=4},s)
```

$\{[s=\sqrt{2}]\}$

Bestimmung des Urbildes des rechten Hauptscheitels:

$$\vec{r}\,''=\begin{pmatrix} r \\ 0 \end{pmatrix},\ \vec{r}\,'=\begin{pmatrix} r \\ 0 \end{pmatrix}-\vec{t},\ \vec{r}=P\begin{pmatrix} r \\ 0 \end{pmatrix}+\vec{m},\ r=r\cdot\overrightarrow{ev_1}+\vec{m}$$

(hier nicht werwendet, Mit ev1 rechts ist hier der normierte 1. Eigenvektor gemeint.

```
r:=sqrt(40): //große (rechte) Halbachse
rur:=r*ev2n+m;
```

$$\begin{pmatrix} -\frac{\sqrt{3}\cdot\sqrt{10}\cdot 2}{3}-2 \\ -\frac{\sqrt{3}\cdot\sqrt{10}\cdot 2}{3}-2 \\ \frac{2\cdot\sqrt{3}\cdot\sqrt{10}}{3}+4 \end{pmatrix}$$

```
mp:=plot::Point3d(m,PointSize=2, PointColor=[0,1,0]):
Op:=plot::Point3d([0,0,0], PointSize=2, PointColor=[1,0,0]):

rurp:=plot::Point3d(rur, PointSize=2, PointColor=[0,1,1]):
rp:=plot::Point3d([0,r,0],PointSize=2, PointColor=[0,0,1]):
ev1urp:=plot::Arrow3d(m,m+5*ev1):
ev2urp:=plot::Arrow3d(m,m+5*ev2):
ev3urp:=plot::Arrow3d(m,m+5*ev3):
tp:=plot::Arrow3d(-t|s=0,[0,0,0],LineColor=[1,0,0]):
plot(quadrikp,mp,Op,rurp,tp,
      ev1urp,ev2urp,ev3urp,PointSize=2,Scaling=Constrained);
```




```
plot(quadrikp,quastrichp, quadrikHp,mp,Op,tp,rp,

ev1urp,ev2urp,ev3urp,PointSize=2,Scaling=Constrained)
```






20
10
z
0
-10
-20
x
-20
20
20
10
0
-10
-20
y